

# SCIENCE & VIE

N° 865

OCTOBRE 1989

● Le laser  
à compression  
temporelle

## LE SU 27 CHASSEUR ACROBATE

● Ce que  
mangent les  
Français

● Métiers  
d'avenir :  
la recherche

130 FB - 5,20 FS - Scan 3.50 - 550 Plac. - 480 Esc - 18.50 Dh - 2.300 Dt - 4200 L - USA NYC \$ 3.75 - RCI : 1370 CFA - 845 CFP

M 2578 - 865 - 18,00 F



3792578018002 08650

# Mathematic's Baby

## INFORMATIQUE AMUSANTE

**E**n cette période de rentrée scolaire, le programme que nous vous proposons ce mois-ci s'adresse aux plus jeunes de nos lecteurs, à savoir ceux pour qui l'université n'est encore qu'un projet lointain puisqu'ils viennent tout juste de franchir le cap du CP ou du CE 1. Pourtant, un très abondant courrier nous demande régulièrement de proposer des programmes à vocation éducative.

Nous vous avons déjà proposé, il y a environ deux ans, le jeu du "Lapin bleu", qui, à son époque, avait été apprécié de bien des parents désireux que leurs enfants se familiarisent avec l'informatique et, surtout, avec la frappe au clavier, qui ne présente que des lettres écrites en majuscules d'imprimerie.

Dans le même ordre d'idée, nous vous proposons maintenant un

petit programme, dont le but est que nos chères têtes blondes réalisent des additions ou des soustractions. Certes, elles resteront relativement simples et, qui plus est, il sera possible à l'enfant de déterminer son niveau de difficulté grâce à deux critères.

En premier lieu, il pourra choisir le type d'opération proposée par l'ordinateur. Addition exclusivement (pour les élèves de CP), soustraction seulement (pour ceux de CE 1), ou mélange de ces deux dernières (pour se familiariser avec la reconnaissance du signe).

En second lieu, le nombre de chiffres maximal que devra comporter le résultat est également

```
10 MODE 0
20 REM *****
30 REM *
40 REM * SYMBOLE BARRE D'OPERATION *
50 REM *
60 REM *****
70 SYMBOL AFTER 127
80 SYMBOL 128,0,0,0,255,255,0,0,0
90 REM *****
100 REM *
110 REM * CHOIX DU TYPE D'OPERATION. *
120 REM *
130 REM *****
140 PRINT "VEUX-TU:":PRINT
150 PRINT "1 DES ADDITIONS":PRINT
160 PRINT "2 DES SOUSTRATIONS":PRINT
170 PRINT "3 OU LES DEUX ?":PRINT
180 PRINT "TON CHOIX ?":PRINT:PRINT
190 LET M$=UPPER$(INKEY$)
200 IF M$="" THEN GOTO 190
210 REM *****
220 REM *
230 REM * CHOIX DU NOMBRE DE CHIFFRES *
240 REM * DE L'OPERATION. *
250 REM *
260 REM *****
270 PRINT "CHOISIS TON NIVEAU:":PRINT
280 PRINT "1,2 ou 3 chiffres."
290 LET K$=INKEY$
300 IF K$="" THEN GOTO 290
310 REM *****
320 REM *
330 REM * PRISE EN COMPTE DU CHOIX *
340 REM *
350 REM *****
360 IF K$="1" THEN LET C=10
370 IF K$="2" THEN LET C=100
380 IF K$="3" THEN LET C=1000
390 LET Z=0
400 REM *****
410 REM *
420 REM * DEPART JEU *
430 REM * POUR 10 PROPOSITIONS *
440 REM *
450 REM *****
460 FOR U=1 TO 10
470 REM *****
480 REM *
490 REM * TIRAGE ALEATOIRE *
500 REM * DES DEUX NOMBRES *

510 REM *
520 REM *****
530 LET A=INT (RND*C)
540 LET B=INT (RND*C)
550 REM *****
560 REM *
570 REM * CONTROLE DE LA VALEUR *
580 REM * DE LEUR SOMME. *
590 REM *
600 REM *****
610 IF A+B>=C THEN GOTO 530
620 REM *****
630 REM *
640 REM * DETERMINATION DU TYPE *
650 REM * D'OPERATION PROPOSEE. *
660 REM *
670 REM *****
680 LET S=0
690 IF M$="2" THEN LET S=1
700 IF M$="3" THEN LET S=INT (RND*2)
710 LET J=1
720 REM *****
730 REM *
740 REM * INVERSION DES DEUX NOMBRES *
750 REM * POUR EVITER QUE LE RESULTAT *
760 REM * SOUSTRACTION SOIT NEGATIF. *
770 REM *
780 REM *****
790 IF S=1 AND A<B THEN LET X=A:LET A=B:LET B=X
800 REM *****
810 REM *
820 REM * AFFICHAGE DE L'OPERATION *
830 REM *
840 REM *****
850 CLS
860 PRINT "Je te propose:"
870 REM *****
880 REM *
890 REM * REBOUCLAGE POUR TROIS ESSAIS *
900 REM *
910 REM *****
920 FOR J=1 TO 3
930 LET R=0
940 GOSUB 1530
950 REM *****
960 REM *
970 REM * CONTROLE RESULTAT ADDITION *
980 REM *
```

programmable de 1 à 3. Notre programme pourra donc être utilisé depuis le plus jeune âge. Notons que quelques pièges, du type 2 + 0, seront régulièrement proposés, mais en aucun cas le nombre de chiffres du résultat, retenues comprises, ne sera supérieur au nombre choisi.

Voyons maintenant l'écriture du programme. En premier lieu, nous demanderons à l'Amstrad de travailler en mode 0 de manière à afficher sur l'écran des caractères aussi grands que possible. Puis, à la ligne 80, nous définirons un symbole graphique qui nous servira, par la suite, à afficher la barre de l'opération. Vient ensuite le choix

du niveau de difficulté. Un premier menu, présenté par les lignes de 140 à 180, permettra de choisir le type d'opération posée. Ce choix sera mémorisé par la variable M\$. De même, la détermination du nombre de chiffres souhaité sera assurée par les lignes de 270 à 300. Ici, K\$ sera utilisée pour sa prise en compte.

En effet, des lignes de 360 à 380, nous affecterons à la variable C la valeur 10, 100 ou 1000 en fonction de K\$; valeur que ne devra jamais atteindre le résultat de l'opération proposée. Nous aurons donc ainsi un résultat comportant un, deux ou trois chiffres. Ces deux variables initialisées, nous pourrons en-

tamer la phase de jeu du programme.

En fait, comme nous désirions qu'il représente 10 opérations différentes pour une "partie", il sera encadré par la boucle FOR...NEXT, formée autour de la variable U (ligne 460 et ligne 1300). Nous commencerons ici par déterminer aléatoirement la valeur des deux nombres (A et B) que nous utiliserons pour poser l'opération, puis nous vérifierons que leur valeur est cohérente par rapport au nombre de chiffres que doit composer le résultat. Un test concernant l'addition sera effectué par la ligne 610; en fonction du choix, la valeur maximale de la somme de A

```

990 REM *****
1000 IF S=1 THEN GOTO 1080
1010 IF A+B=R THEN LOCATE 8,12:PRINT "BRA
VO !":Z=Z+1:LET J=3:GOTO 1230
1020 IF S=0 THEN GOTO 1090
1030 REM *****
1040 REM *
1050 REM *   CONTROLE R. SOUSTRACTION *
1060 REM *
1070 REM *****
1080 IF A-B=R THEN LOCATE 8,12:PRINT "BRA
VO !":Z=Z+1:LET J=3:GOTO 1230
1090 LOCATE 3,12:PRINT "REFLECHIS BIEN;"
1100 LOCATE 3,14:PRINT "TU T'ES TROMPE!"
1110 IF J<>4 THEN LOCATE 5,9:PRINT " "
1120 REM *****
1130 REM *
1140 REM *   REBOUCLAGE SUR ERREUR *
1150 REM *
1160 REM *****
1170 NEXT J
1180 REM *****
1190 REM *
1200 REM *   BOUCLE DE TEMPORISATION *
1210 REM *
1220 REM *****
1230 FOR T=1 TO 2000:NEXT T
1240 REM *****
1250 REM *
1260 REM *   OPERATION SUIVANTE. *
1270 REM *
1280 REM *****
1290 CLS
1300 NEXT U
1310 REM *****
1320 REM *
1330 REM *   AFFICHAGE DE LA NOTE OBTENUE *
1340 REM *
1350 REM *****
1360 LOCATE 1,4:PRINT "TA NOTE: ";Z;"/ 10"
1370 REM *****
1380 REM *
1390 REM *   DEMANDE DE NOUVELLE SERIE. *
1400 REM *
1410 REM *****
1420 LOCATE 3,12:PRINT "POUR UN AUTRE "
1430 LOCATE 3,14:PRINT "ESSAI TAPE "
1440 LOCATE 3,16:PRINT "SUR LA TOUCHE 'P'
"
1450 IF UPPER$(INKEY$)<>"P" THEN GOTO 145
0
1460 GOTO 10
1470 REM *****
1480 REM *
1490 REM *   SOUS-ROUTINE D'AFFICHAGE *
1500 REM *   DES CHIFFRES FRAPPES. *
1510 REM *
1520 REM *****
1530 IF J=1 THEN GOSUB 1730
1540 IF S=0 THEN LOCATE 4,6:PRINT "+"
1550 IF S=1 THEN LOCATE 4,6:PRINT "-"
1560 FOR I=0 TO 2-ST
1570 LOCATE 8-I,9:PEN 14:PRINT "*":PEN 1
1580 LET K$=INKEY$: IF K$="" THEN GOTO 15
80
1590 LET K=ASC(K$)-48
1600 IF K<0 OR K>9 THEN GOTO 1580
1610 LOCATE 8-I,9:PRINT K$
1620 LET R=R+(K*10^I)
1630 LOCATE 3,12:PRINT " "
1640 LOCATE 3,14:PRINT " "
1650 NEXT I
1660 RETURN
1670 REM *****
1680 REM *
1690 REM *   SOUS-ROUTINE D'AFFICHAGE DE *
1700 REM *   L'OPERATION ET COMMENTAIRES *
1710 REM *
1720 REM *****
1730 LET E=2
1740 IF A>9 THEN LET E=1
1750 IF A>100 THEN LET E=0
1760 LOCATE 5+E,5:PRINT A
1770 LET ST=E
1780 LET E=2
1790 IF B>9 THEN LET E=1
1800 IF B>100 THEN LET E=0
1810 LOCATE 5+E,7:PRINT B
1820 IF ST>E THEN LET ST=E
1830 FOR I=0 TO 5
1840 LOCATE (4+I),8:PRINT CHR$(128)
1850 NEXT I
1860 LOCATE 1,18:PRINT "A TOI DE JOUER:"
1870 LOCATE 1,20:PRINT "INDIQUE LE CHIFFR
E"
1880 LOCATE 1,22:PRINT "QUI DOIT REMPLACE
R"
1890 LOCATE 1,24:PRINT "L'ETOILE: ";:PEN 1
4:PRINT " *":PEN 1
1900 RETURN

```

et B ne devra pas être supérieure à 9, 99 ou 999. Au cas contraire, un nouveau tirage aléatoire sera effectué. De même, dans le cas d'une soustraction, nous intervertirons les valeurs de A et B si le résultat obtenu se trouvait avoir une valeur négative (ligne 790). Les lignes de 680 à 700 feront, quant à elles, office d'aiguillage en fonction du type d'opération à proposer. Vient alors l'affichage de l'opération ainsi déterminée. Pour cela, nous utiliserons une sous-routine que nous étudierons plus loin. Notons cependant que, en cas d'erreur sur la réponse fournie, trois essais successifs seront proposés.

Après chaque réponse fournie, un contrôle sera effectué. Il sera réalisé par la ligne 1010 pour une addition et par la ligne 1080 pour une soustraction. Enfin, si le résultat est exact, le programme passera à l'option suivante. Une fois les 10 problèmes proposés, l'ordinateur indiquera la note obtenue (ligne 1360), puis demandera à l'enfant s'il désire faire une nouvelle "partie" ou non.

Notre programme comporte deux sous-routines. La première, comprise entre les lignes 1530 et 1660, permet, d'une part, l'affichage sous la ligne de résultat de chaque chiffre tapé au clavier et, d'autre part, indique, grâce à une étoile, l'emplacement que doit occuper le chiffre à frapper. La seconde, lignes de 1730 à 1900, a pour mission d'afficher sur l'écran l'opération proposée.

La frappe de ce programme ne doit pas poser de problème particulier. Certes, il est conçu pour l'Amstrad CPC, mais il reste parfaitement adaptable à d'autres machines. Précisons que MODE 0 peut être ignoré. Les instructions SYMBOL AFTER et SYMBOL permettent de redéfinir un symbole graphique. De même, LOCATE permet d'effectuer un positionnement du curseur et peut être remplacé par HTAB, VTAB ou PRINT AT, etc. Enfin, PEN permet de choisir la couleur, ou le clignotement, des caractères affichés ; il a pour équivalent, sur le MO 5 par exemple, SCREEN ou, sur d'autres machines, INK. Le reste des instructions basic utilisées demeure en revanche parfaitement standard et peut donc être directement exploité.

**Henri-Pierre Penel**